

```

function shp2str,file
;+
; Procedure to convert shapefile attribute table to IDL structure

; Created 5/22/06 HAZ

; Syntax: Result=shp2str([shapefilename])
;
; Returns anonymous structure with all fields from attribute table
; [shapefilename] optional - otherwise prompts user
;
; Issues:
; Blank values are converted to zeros or null strings
; On error, returns -1
;-

; Check for file
n=n_params()
if n lt 1 then begin
  file=dialog_pickfile(/READ,PATH='d:\',FILTER="*.shp",/FIX_FILTER,/MUST_EXIST)
endif else begin
  if not(FILE_TEST(file)) then begin
    print, 'SHP2STR: ***ERROR - file does not exist'
    return,-1
  endif
  if STRLOWCASE(STRMID(file,3,4,/REVERSE_OFFSET)) ne '.shp' then begin
    print, 'SHP2STR: ***ERROR - file not shapefile'
    return,-1
  endif
endelse

myshape=OBJ_NEW('IDLffShape',file)
myshape->IDLffShape::GetProperty,ATTRIBUTE_NAMES=anames
attr=myshape->IDLffShape::GetAttributes(/ALL)

if(n_elements(unique(anames,sort(anames))) ne n_elements(anames)) then begin
  print, 'SHP2STR: ***ERROR - field names are not unique'
  OBJ_DESTROY, myshape
  return,-1
endif

; transfer to structure with field names
att=CREATE_STRUCT(anames[0],attr.(0))
for i=1,n_elements(anames)-1 do begin
  att=CREATE_STRUCT(att,anames[i],attr.(i))
endfor

; Close the shapefile.
OBJ_DESTROY, myshape

return,att

```

END